<commit_before>{stdenv, fetchFromGitHub
  , llvmPackages
  , cmake, boehmgc, gmp, zlib, ncurses, boost
  , python, git, sbcl
}:
stdenv.mkDerivation rec {
  name = "${pname}-${version}";
  pname = "clasp";
  version = "0.4.99.20170801";

  src = fetchFromGitHub {
    owner = "drmeister";
    repo = "clasp";
    rev = "525ce1cffff39311e3e7df6d0b71fa267779bdf5";
    sha256 = "1jqya04wybgxnski341p5sycy2gysxad0s5q8d59z0f6ckj3v8k1";
    fetchSubmodules = true;
  };

  nativeBuildInputs = [ cmake python git sbcl ];

  buildInputs = with llvmPackages; (
    builtins.map (x: stdenv.lib.overrideDerivation x
         (x: {NIX_CFLAGS_COMPILE= (x.NIX_CFLAGS_COMPILE or "") + " -frtti"; }))
      [ llvm clang clang-unwrapped clang ]) ++
  [
    gmp zlib ncurses
    boost boehmgc
    (boost.override {enableStatic = true; enableShared = false;})
    (stdenv.lib.overrideDerivation boehmgc
      (x: {configureFlags = (x.configureFlags or []) ++ ["--enable-static"];}))
  ];

  NIX_CFLAGS_COMPILE = " -frtti ";

  configurePhase = ''
    runHook preConfigure

    export CXX=clang++
    export CC=clang

    echo "
      INSTALL_PATH_PREFIX = '$out'
    " | sed -e 's/^ *//' > wscript.config

    python ./waf configure update_submodules

    runHook postConfigure
  '';

  buildPhase = ''
    runHook preBuild

    python ./waf build_cboehm

    runHook postBuild
  '';

  installPhase = ''
    runHook preInstall

    python ./waf install_cboehm

    runHook postInstall
  '';

  meta = {
    inherit version;
    description = ''A Common Lisp implementation based on LLVM with C++ integration'';
    license = stdenv.lib.licenses.lgpl21Plus ;
    maintainers = [stdenv.lib.maintainers.raskin];
    platforms = stdenv.lib.platforms.linux;
    homepage = "https://github.com/drmeister/clasp";
  };
}<commit_msg>clasp-common-lisp: mark as broken<commit_after>{stdenv, fetchFromGitHub
  , llvmPackages
  , cmake, boehmgc, gmp, zlib, ncurses, boost
  , python, git, sbcl
}:
stdenv.mkDerivation rec {
  name = "${pname}-${version}";
  pname = "clasp";
  version = "0.4.99.20170801";

  src = fetchFromGitHub {
    owner = "drmeister";
    repo = "clasp";
    rev = "525ce1cffff39311e3e7df6d0b71fa267779bdf5";
    sha256 = "1jqya04wybgxnski341p5sycy2gysxad0s5q8d59z0f6ckj3v8k1";
    fetchSubmodules = true;
  };

  nativeBuildInputs = [ cmake python git sbcl ];

  buildInputs = with llvmPackages; (
    builtins.map (x: stdenv.lib.overrideDerivation x
         (x: {NIX_CFLAGS_COMPILE= (x.NIX_CFLAGS_COMPILE or "") + " -frtti"; }))
      [ llvm clang clang-unwrapped clang ]) ++
  [
    gmp zlib ncurses
    boost boehmgc
    (boost.override {enableStatic = true; enableShared = false;})
    (stdenv.lib.overrideDerivation boehmgc
      (x: {configureFlags = (x.configureFlags or []) ++ ["--enable-static"];}))
  ];

  NIX_CFLAGS_COMPILE = " -frtti ";

  configurePhase = ''
    runHook preConfigure

    export CXX=clang++
    export CC=clang

    echo "
      INSTALL_PATH_PREFIX = '$out'
    " | sed -e 's/^ *//' > wscript.config

    python ./waf configure update_submodules

    runHook postConfigure
  '';

  buildPhase = ''
    runHook preBuild

    python ./waf build_cboehm

    runHook postBuild
  '';

  installPhase = ''
    runHook preInstall

    python ./waf install_cboehm

    runHook postInstall
  '';

  meta = {
    inherit version;
    description = ''A Common Lisp implementation based on LLVM with C++ integration'';
    license = stdenv.lib.licenses.lgpl21Plus ;
    maintainers = [stdenv.lib.maintainers.raskin];
    platforms = stdenv.lib.platforms.linux;
    homepage = "https://github.com/drmeister/clasp";
    broken = true; # 2018-09-08, no successful build since 2018-01-03
  };
}
{'fraction_non_alphanumeric': 0.12506753106428958, 'fraction_numerical': 0.03511615343057806, 'mean_word_length': 3.123608017817372, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 5, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 2, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6192247', 'n_tokens_mistral': 1461, 'n_tokens_neox': 1377, 'n_words': 271}